# Maintainer: Andreas Radke <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-appres
pkgver=1.0.5
pkgrel=2
pkgdesc="List X application resource database"
arch=('x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxt')
makedepends=('xorg-util-macros')
conflicts=('xorg-res-utils')
source=(https://xorg.freedesktop.org/archive/individual/app/appres-${pkgver}.tar.bz2{,.sig})
sha512sums=('70825563cd80f0f433c48ab2fdae888072939c1b22fb22340b70b9444c97916275ed92c9e5bf9a06bb76c32a5371b89ff7c0211a9a47a227c88311078c2c4137'
            'SKIP')
validpgpkeys+=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd "${srcdir}/appres-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/appres-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
